NAG Library Routine Document

F16ECF (BLAS_DAXPBY)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F16ECF (BLAS_DAXPBY) computes the sum of two scaled vectors, for real vectors and scalars.

2 Specification

```
SUBROUTINE F16ECF (N, ALPHA, X, INCX, BETA, Y, INCY)

INTEGER            N, INCX, INCY
REAL (KIND=nag_wp) ALPHA, X(1+(N-1)*ABS(INCX)), BETA, Y(1+(N-1)*ABS(INCY))
```

The routine may be called by its BLAST name ***blas_daxpby***.

3 Description

F16ECF (BLAS_DAXPBY) performs the operation

$$y \leftarrow \alpha x + \beta y$$

where x and y are n-element real vectors, and α and β real scalars. If n is less than or equal to zero, or if α is equal to zero and β is equal to 1, this routine returns immediately.

4 References

Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001) *Basic Linear Algebra Subprograms Technical (BLAST) Forum Standard* University of Tennessee, Knoxville, Tennessee http://www.netlib.org/blas/blast-forum/blas-report.pdf

5 Parameters

1: N – INTEGER *Input*

On entry: n, the number of elements in x and y.

2: ALPHA – REAL (KIND=nag_wp) *Input*

On entry: the scalar α.

3: X$(1 + (\text{N} - 1) \times |\text{INCX}|)$ – REAL (KIND=nag_wp) array *Input*

On entry: the n-element vector x.

If INCX > 0, x_i must be stored in X$(1 + (i - 1) \times \text{INCX})$, for $i = 1, 2, \ldots, \text{N}$.

If INCX < 0, x_i must be stored in X$(1 - (\text{N} - i) \times \text{INCX})$, for $i = 1, 2, \ldots, \text{N}$.

Intermediate elements of X are not referenced.

4: INCX – INTEGER *Input*

On entry: the increment in the subscripts of X between successive elements of x.

Constraint: INCX $\neq 0$.

5: BETA – REAL (KIND=nag_wp) *Input*

On entry: the scalar β.

6: Y$(1 + (\mathrm{N} - 1) \times |\mathrm{INCY}|)$ – REAL (KIND=nag_wp) array *Input/Output*

On entry: the n-element vector y.

If INCY > 0, y_i must be stored in Y$(1 + (i - 1) \times \mathrm{INCY})$, for $i = 1, 2, \ldots, \mathrm{N}$.

If INCY < 0, y_i must be stored in Y$(1 - (\mathrm{N} - i) \times \mathrm{INCY})$, for $i = 1, 2, \ldots, \mathrm{N}$.

Intermediate elements of Y are not referenced.

On exit: the updated vector y stored in the array elements used to supply the original vector y.

Intermediate elements of Y are unchanged.

7: INCY – INTEGER *Input*

On entry: the increment in the subscripts of Y between successive elements of y.

Constraint: INCY $\neq 0$.

6 Error Indicators and Warnings

IFAIL =

On entry, INCX = ⟨*value*⟩.
Constraint: INCX $\neq 0$.

On entry, INCY = ⟨*value*⟩.
Constraint: INCY $\neq 0$.

On entry, N = ⟨*value*⟩.
Constraint: N ≥ 0.

7 Accuracy

The BLAS standard requires accurate implementations which avoid unnecessary over/underflow (see Section 2.7 of Basic Linear Algebra Subprograms Technical (BLAST) Forum (2001)).

8 Further Comments

None.

9 Example

This example computes the result of a scaled vector accumulation for

$$\alpha = 3, \qquad x = (-4, 2.1, 3.7, 4.5, -6)^{\mathrm{T}},$$
$$\beta = -1, \qquad y = (-3, -2.4, 6.4, -5, -5.1)^{\mathrm{T}}.$$

9.1 Program Text

```
    Program f16ecfe

!     F16ECF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: blas_daxpby, nag_wp
!     .. Implicit None Statement ..
      Implicit None
```

```
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: alpha, beta
      Integer                          :: incx, incy, n, nx, ny
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: x(:), y(:)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: abs
!     .. Executable Statements ..
      Write (nout,*) 'F16ECF Example Program Results'

!     Skip heading in data file
      Read (nin,*)

      Read (nin,*) n
      Read (nin,*) incx, incy

      nx = 1 + (n-1)*abs(incx)
      ny = 1 + (n-1)*abs(incy)
      Allocate (x(nx),y(ny))

      Read (nin,*) alpha, beta
      Read (nin,*) x(1:nx:abs(incx))
      Read (nin,*) y(1:ny:abs(incy))

!     Compute Y = ALPHA*X + BETA*Y

      Call blas_daxpby(n,alpha,x,incx,beta,y,incy)

      Write (nout,*)
      Write (nout,99999)
      Write (nout,99998) y(1:ny:abs(incy))

99999 Format (1X,'Result of scaled vector addition is')
99998 Format (1X,'Y =',5F9.4)
    End Program f16ecfe
```

9.2 Program Data

```
F16ECF Example Program Data
   5                                  : n
  -1    -1                            : incx and incy
   3.0  -1.0                          : alpha and beta
  -4.0   2.1   3.7   4.5   -6.0       : x
  -3.   -2.4   6.4  -5.0   -5.1       : y
```

9.3 Program Results

```
 F16ECF Example Program Results

 Result of scaled vector addition is
 Y =  -9.0000   8.7000   4.7000  18.5000 -12.9000
```